

```

10 CLS: REV$ = "$Header: d:/protocol/p/RCS/pproto.bas 1.3 2000-11-21 07:41:57-08 Hamilton Exp Hamilton $"
20 PRINT REV$
30 PRINT "Example P-Protocol Commands"
40 PRINT "Unsupported software from Pelco"
50 '
60 OPEN "COM1:4800,N,8,1" AS #2
70 CMNDCOUNT = 0
80 PANSPEED = 32
90 TILTSPEED = 32
100 SOM = &HA0: REM Start of message
110 EOM = &HAF: REM End of message
120 ADDRESS = 2
130 CMND1 = 0
140 CMND2 = 0
150 DATA1 = 0
160 DATA2 = 0
170 '
180 REM Get direction to move the dome
190 KPAD$ = INKEY$
200 IF LEN(KPAD$) = 0 THEN 180
210 IF KPAD$ = "x" OR KPAD$ = "X" THEN 750
220 IF KPAD$ = "s" OR KPAD$ = "S" THEN 390
230 IF KPAD$ = "a" OR KPAD$ = "A" THEN 480
240 IF LEN(KPAD$) = 1 THEN BYTE3 = ASC(KPAD$): GOTO 560
250 IF LEN(KPAD$) <> 2 THEN 180
260 BYTE3 = ASC(MID$(KPAD$,2,1))
270 '
280 REM &Hxx is used to indicate hex numbers
290 IF BYTE3 = &H4D THEN CMND2 = &H2: DATA1 = PANSPEED: ID$ = "R": GOTO 560: REM Right
300 IF BYTE3 = &H4B THEN CMND2 = &H4: DATA1 = PANSPEED: ID$ = "L": GOTO 560: REM Left
310 IF BYTE3 = &H48 THEN CMND2 = &H8: DATA2 = TILTSPEED: ID$ = "U": GOTO 560: REM Up
320 IF BYTE3 = &H50 THEN CMND2 = &H10: DATA2 = TILTSPEED: ID$ = "D": GOTO 560: REM Down
330 IF BYTE3 = &H47 THEN CMND2 = &H40: ID$ = "W": GOTO 560: REM Wide angle
340 IF BYTE3 = &H49 THEN CMND2 = &H20: ID$ = "T": GOTO 560: REM Tele
350 REM Woops it's not a used code, print it out and try again
360 PRINT "";"HEX$(BYTE3);"";
370 GOTO 180
380 '
390 REM Ask for a new speed
400 INPUT "Enter in a new positive speed less than 64 (cr = 32)";SPEED
410 IF SPEED = 0 THEN SPEED = 32
420 IF SPEED > 64 OR SPEED < 0 THEN 400
430 PANSPEED = SPEED
440 TILTSPEED = SPEED: IF SPEED = 64 THEN TILTSPEED = 63
450 PRINT "Using pan/tilt speeds of:";PANSPEED;"/";TILTSPEED
460 GOTO 180
470 '
480 REM Ask for a new camera address
490 REM Note that camera addresses are one based
500 INPUT "What camera address should be used (cr = camera 2)";ADDRESS
510 IF ADDRESS = 0 THEN ADDRESS = 2
520 IF ADDRESS > 255 OR ADDRESS < 0 THEN 500
530 PRINT "Using a camera address of:";ADDRESS
540 GOTO 180
550 '
560 REM Send it out
570 PRINT USING"\\ ##: ";ID$,CMNDCOUNT;
580 CMNDCOUNT = CMNDCOUNT + 1: IF CMNDCOUNT > 99 THEN CMNDCOUNT = 0

```

```

590 GOSUB 930: REM Send it
600 '
610 REM For next time start with everything at zero
620 CMND1 = 0
630 CMND2 = 0
640 DATA1 = 0
650 DATA2 = 0
660 '
670 REM Now stop it moving after a second, or more, of motion
680 REM Note that the subroutines may move if this gets renumbered and
690 REM the next few lines are commented out.
700 REM GOSUB 770: REM Delay a sec
710 REM GOSUB 830: REM Send it
720 REM GOSUB 770: REM Delay a sec
730 GOTO 180
740 '
750 REM All done
760 PRINT "All done"
770 PRINT DATE$,TIME$
780 END
790 '
800 REM Wait for up to one second
810 X$ = RIGHT$(TIME$,2)
820 IF X$ = RIGHT$(TIME$,2) GOTO 820
830 RETURN
840 '
850 REM Get the checksum
860 CHECKSUM = SOM XOR EOM: REM Do these first
870 CHECKSUM = CHECKSUM XOR ADDRESS: REM Now XOR in the address
880 CHECKSUM = (CHECKSUM XOR CMND1) XOR CMND2: REM And the commands
890 CHECKSUM = (CHECKSUM XOR DATA1) XOR DATA2: REM Then the data
900 CHECKSUM = CHECKSUM AND 255
910 RETURN
920 '
930 REM Send it out and display it
940 GOSUB 850: REM Get checksum
950 D1$ = CHR$(SOM)
960 D2$ = CHR$(ADDRESS)
970 D3$ = CHR$(CMND1)
980 D4$ = CHR$(CMND2)
990 D5$ = CHR$(DATA1)
1000 D6$ = CHR$(DATA2)
1010 D7$ = CHR$(EOM)
1020 D8$ = CHR$(CHECKSUM)
1030 PRINT #2,D1$;D2$;D3$;D4$;D5$;D6$;D7$;D8$;
1040 X = SOM: GOSUB 1150
1050 X = ADDRESS: GOSUB 1150
1060 X = CMND1: GOSUB 1150
1070 X = CMND2: GOSUB 1150
1080 X = DATA1: GOSUB 1150
1090 X = DATA2: GOSUB 1150
1100 X = EOM: GOSUB 1150
1110 X = CHECKSUM: GOSUB 1150
1120 PRINT
1130 RETURN
1140 '
1150 REM Print out a command byte
1160 X$ = HEX$(X)

```

```
1170 IF LEN(X$) = 1 THEN X$ = "0" + X$  
1180 PRINT X$;" ";  
1190 RETURN
```